Not to be cited without
authors' permission

CAFSAC Res. Doc. 78/30

A model and computer program used to assess yield per recruit in Newfoundland lobster stocks

by

G. P. Ennis and S. A. Akenhead

Dept. of Fisheries and Oceans
Fisheries and Marine Service
Research and Resource Services Branch
3 Water Street
St. John's, Newfoundland
A1C 1A1

Introduction

In the absence of effective effort control, the minimum size regulation is probably the best management tool available to prevent overexploitation of lobster stocks. Provided such a regulation can be enforced, yields from the resource can be maximized by selecting the appropriate size limit for current exploitation rates. If effective effort control is feasible, the appropriate exploitation rate for a particular size limit can likewise be selected. In order to determine the "appropriate" size limit or exploitation rate a yield per recruit assessment is necessary. For those species in which age determiniation is possible and for which an estimate of natural mortality is available, such an assessment can be done quite readily using the well established Beverton and Holt model. In the case of species such as the lobster where aging is not possible, a yield per recruit assessment is not so readily done unless some other method can be used to estimate the parameters of the von Bertalanffy growth equation. If growth is described in some form other than the von Bertalanffy equation, the yield per recruit model used will have to be in terms of the particular growth functions available. One approach being used to estimate growth rates in Newfoundland populations of lobsters is to combine data on molt increment and proportions molting. This paper describes a yield per recruit model for lobsters which uses these growth functions and a computer program which does the analysis.

Biological Basis of the Model

The model considers males and females separately and for each starts with an initial population of 1000 animals which are distributed evenly over the 10 one mm size groups from 60 to 69 mm carapace length. The assumed even distribution over this size range was compared (by chi-square) with an observed frequency distribution from diver caught samples. In the case of males P > .3 and for females P > .9.

A carapace length-whole weight relationship in the form $\log_{10} Y = a + b \log_{10} X$ is used. This relationship is obtained from extensive biological sampling (see Ennis 1971). Molt increment or more precisely postmolt carapace length is derived from the premolt-postmolt relationship ($y = a + bx$) obtained from "sphyrion" tagging (see Ennis 1972, 1978). The distribution of postmolt lengths around the estimated mean postmolt length for any premolt length is derived from the covariance of premolt and postmolt lengths using a table of the normal distribution. Proportion molting is derived from the probit analysis of number molted and total number examined at each carapace length as described in Ennis (1978).

The reason for selecting 60 mm carapace length as the smallest size to be considered is too few observations on shell condition and molt increment are available for smaller sizes. No existing size limits for this species are below 60 mm and it is well below the existing size limit (81 mm) for Newfoundland stocks. The maximum size to be considered was arbitrarily chosen at 149 mm giving 90 one mm size groups over the range of sizes considered.

No direct estimates of natural mortality in lobsters are available but the general consensus reached by the ICES Working Group on <u>Homarus</u> Stocks is that it can be expected to be less than 10% annually (Anon., 1977) One would expect most natural mortality in lobsters to be associated with molting and in this model the full 10% natural mortality rate is applied only to those lobsters that molt. Lobsters not molting in a particular year are subjected to a 5% rate of natural mortality. Since proportion molting decreases with increasing size, effectively a decreasing rate of natural mortality (10 to 5%) is applied as size increases.

Special treatment is required for females because of egg-laying and the fact that egg-bearing females are protected from the fishery. The percentage of non-egg-bearing females present in a population in early summer that lay eggs during the spawning season later in the summer varies with size. One "sphyrion" tagging project gave the following results for non-egg-bearing females tagged prior to the molting and spawning season and recaptured afterwards but before the start of the next year's molting and spawning season.

Carapace length at tagging	No. non-egg-bearing females tagged	% egg-bearing when recaptured
66-70	6	16.7
71-75	13	38.5
76-80	22	81.8
81-85	15	60.0
86-90	8	62.5
91-125	5	40.0

In the same tagging project all of 69 egg-bearing females ranging in carapace length from 68 to 106 mm tagged prior to the molting and spawning season were non-egg-bearing and had molted when recaptured afterwards.

A problem arose at the smaller sizes in that when the proportion molting and proportion laying eggs were totaled it equalled more than 100%. A very small number of females molt and lay eggs in the same season but not enough to account for the discrepancy. It was decided to multiply the number of females at a particular size by the proportion molting for that size first then assume that all females that did not molt would lay eggs. These are protected from exploitation for the year they are berried but are subjected to 5% natural mortality. It is assumed that females laying eggs in one year all molt the following year and are subjected to 10% natural mortality in the year that they molt.

Using the foregoing biological information, yield per recruit values are produced as described in the following section for rates of exploitation from 20 to 100% (in 10% increments) at recruitment lengths (ie. size limits) of 70 mm (2 3/4"), 76 mm (3"), 81 mm (3 3/16"), 89 mm (3 1/2"), 95 mm (3 3/4") and 102 mm (4").

Mechanics of the Program

The integration of the component functions described for growth could only be done by numerical techniques because of the "either bears eggs or molts" formulation which confounds the growth and mortality, and because of the discontinuous nature of the fishing mortality, and lastly because it was desired to simulate the natural variability in growth. This led to the simulation model approach of following a cohort of lobsters through their lives (or until they were very rare), accounting for growth and growth variability, natural mortality as described, catches, and the numbers of berried females. Since males do not have the protection offered by being berried, and also do not have the complication of egg-bearing in their molting sequence, essentially a second model for males was written as a subset of the calculations for females.

To summarize this simulation model of lobster yield per recruit in mathematical terms, it is a numerical double integration over length and time, using continuous functions to control discrete distributions of lengths, and independent natural and controllable mortalities in a complex life-history pattern. The following sections refer to these aspects of the model one by one.

Mortality

From the perspective of modelling, a convenient feature of lobster ecology is the shortness of the fishing season and its separation from the molting season. This allows all the mortalities to be handled separately and additively without need of respecting compensatory mortality. The annual cycle of modelled features is represented as Fig. 1. Notice how females leave the cycle of males for a year while they are berried.

Growth

Growth is handled in terms of millimeter size categories, referring to carapace length, from 60 to 149 mm. All the molting animals in a particular size category are distributed into a distribution of new size categories according to the premolt-postmolt regression, and the deviations around this regression. If, however, some animals do not molt, then either they are not removed from the size category for redistribution, or they are transferred to a holding array for a year while they avoid fishing and growth, but suffer hard-shelled mortality. Figure 2 names the relevant arrays in the program and shows the relationship between the discrete procedures of the model and the continuous function derived from the biological studies.

A complication arises from the "either-or" restriction currently being applied to molting and berrying of females. Since the consequence of berrying is to molt next year and to escape fishing mortality next year, survivorship is high for females, and in those older groups where the proportion molting is low in nature, the model forces higher molting and consequently higher growth. This effect is small at high fishing mortalities, which are normally encountered in the fishery. However, it must be noted that at low fishing mortality rates, the yield per recruit for females is biased toward being too high, mainly due to better growth than observed in old females.

Fishing

Catch rates are applied to the vector of numbers at length as an absolute survivorship (i.e. percent taken in the fishing season), and applied independently of other mortalities as explained before. The nature of a lobster trap and lobster fishing is such that knife-edge selection is a suitable description of the pattern of catchability at length. The two basic parameters of yield per recruit analysis that are available as management variables are thus handled very simply by the model. Notice that the absolute fishing rate is related to the exponential of the instantaneous rate of fishing.

Model output

A wide variety of special outputs can be derived from the model, although for most operations a very terse output is satisfactory. Distributions of length at age for a cohort subject to different exploitation and growth parameters, annual trends in mean size, number of berried females produced through the cohort's history, and matrices of yield valves (catch weight or mean size; free population numbers, weight, or mean size) are all easily obtained by just identifying and printing the appropriate variables in the model. A copy of the program is appended.

REFERENCES

Anon. 1977. Report of the Working Group on Homarus Stocks. ICES CM 1977/K:11

Ennis, G. P. 1971. Lobster (Homarus americanus) fishery and biology in Bonavista Bay, Newfoundland 1966 -70. Fish. Res. Board Can., Tech. Rep. No. 289, 46 p.

Ennis, G. P. 1972. Growth per moult of tagged lobsters (_Homarus americanus_) in Bonavista Bay, Newfoundland. J. Fish. Res. Board Can. 29: 143-148.

1978. Growth curves for Newfoundland lobsters from data on molt increment and proportion molting. CAFSAC Res. Doc. 78/29



Fig. 1. A conceptual summary of the annual cycle of Newfoundland lobsters and the relationships of various mortalities.



Fig. 2. The correspondence between residuals around the growth regression and the discrete handling of growth in the computer model.

APPENDIX I

```
//H7013503 JOB (7013,503C,1,3),ENNIS,CLASS=C
//SCOTT EXEC FORTGCLG,PARM=NOMAP
//FORT.SYSIN DD *
C                                                                           16
C                                                                           22
C                 LOBSTER  YIELD PER RECRUIT ANALYSIS
C                                          SCOTT AKENHEAD  AUG.77
C     LAST UPDATE: MARCH 7 78
C
C     INPUT
C     CARD 1          TITLE CARD    (20A4)
C                COLS 77-80 ARE '   F' FOR RUN OF FEMALES. ANYTHING ELSE
C                IS A RUN FOR MALES.
C
C     CARD 2          RUN  PARAMETERS (10F8.0)
C                START LENGTH, LAST LENGTH, MOULTING SURVIVORSHIP, HARD-SHELLED
C                ANNUAL SURVIVORSHIP, A AND B  OF LENGTH-WEIGHT REGRESSION,
C                A AND B OF PRE-POST-MOULT GROWTH REGRESSION (IN LOG10).
C
C     CARD 3          GROWTH  SCATTER (10F8.0)
C                DIST. AROUND MEAN PREDICTED POSTMOULT LENGTH, FROM 5-4,
C                4-3,3-2,2-1,1-0,0-1,1-2,2-3,3-4,4-5.   IN PROPORTION PER
C                MILLIMETER.
C
C     CARD 4          FISHING PATTERN (10F8.0)
C                FISHING RATE FIRST, FISHING RATE LAST, INCREMENTS OF
C                FISHING RATE, UP TO 7 VALUES OF RECRUIT LENGTH IN MM.
C
C
C     CARD 5         PROPORTION MOULTING  (10F8.0)
C                FROM SMALLEST, ONE FOR EACH LENGTH CLASS, SEVERAL CARDS.
C
C     CARD 6         INITIAL POPULATION   (10F8.0)
C                RECRUIT LENGTH DISTRIBUTION, USE OVER 1000 ANIMALS.
C                AS MANY VALUES AS THERE ARE LENGTH EXPECTED. FOR FULL
C                SIZE RUN, THIS IS NINE CARDS OF BOTH 5 AND 6.
C
C     *******************************************************************
C
C
C
C
      INTEGER  IGROW(90),UNDERL(30)/30*'____'/,LIST(20)                  1
      INTEGER F/'   F'/
       REAL RLVEC(6)/6*0./
      REAL POP(90)/90*0.0/,HOLD(90)/90*0.0/,PRMOLT(90),CAT(90),W(90)     2
       REAL PINIT(90)/90*0./
      REAL EGGERS(90)/90*0./,GDIST(10)/10*0./
C
C   *********************************************************************
C
C      SKIP INITIALIZATIONS FIRST TIME THROUGH
      GOTO 304
C
C          ***   INITIALIZE STORAGE FOR RECYCLING ON NEXT DATASET.
C
300    DO 301 I=1,90
      POP(I)=0.
      HOLD(I)=0.
      PINIT(I)=0.
301    EGGERS(I)=0.
      DO 302 I=1,10
302    GDIST(I)=0.
      DO 303 I=1,6
303      RLVEC(I)=0.
304    CONTINUE
C
C
C     READ CARD 1     COMMENTS ON TITLE CARD
C         AN F IN COLUMN 80 SPECIFIES A RUN FOR FEMALES. ANYTHING ELSE
C         IS A RUN FOR MALES.
C
      READ(5,2,END=305)LIST
      GOTO 306
305    STOP
306      CONTINUE
C
```

```
      ISEX=1
      IF(LIST(20).EQ.    F)ISEX=2
C
C     READ  CARD 2
      READ(5,1) AS,AF,AMORT,BMORT,ALW,BLW,AGM,BGM
      NL=AF-AS
      NL=NL+1
      LS=AS
      LF=AF
      WRITE(6,13)LS,LF,NL,AMORT,BMORT,ALW,BLW,AGM,BGM
      LS=LS-1
C
C     READ PROP. OF POP. FALLING INTO +/- 5MM OF MEAN POSTMOULT LENGTH.
C     THIS VECTOR IS CALCULATED FROM STD.DEV. OF GROWTH REGRESSION.
C
C      READ CARD 2.1  RESIDUAL GROWTH VARIATION
C
      READ(5,1)GDIST
      WRITE(6,14)GDIST
C
C      READ CARD 2.2  CONTROLS FOR F AND RECRUIT LENGTH
C
      READ(5,1)FSTRT,FSTOP,FINCR,RLVEC
       WRITE(6,15)FSTRT,FSTOP,FINCR,RLVEC
C
C     CARD    3   PROPORTION AT EACH LENGTH MOULTING AND EXPOSED TO NAT.
C                 MORTALITY EACH YEAR.
      READ(5,1) (PRMOLT(I),I=1,NL)
C
C     CARD    4   INITIAL POP. DIST'N
C
C
      READ(5,1)(PINIT(I),I=1,NL)
20    CONTINUE
C
C     CALCULATE VECTOR OF WEIGHTS FOR LENGTHS
C
C     TEST PRINT DATA
C         HEADER
      WRITE(6,4)
C         DATA IN COLUMNS
      ALW=10.**ALW
      STARTN=0.
      DO 11   I=1,NL
      W(I)=ALW*FLOAT(LS+I)**BLW
      IGROW(I)=AGM + BGM * FLOAT(LS+I)
      IL=LS+I
      WRITE(6,3) IL,PINIT(I),PRMOLT(I),IGROW(I),W(I)
      STARTN=STARTN+PINIT(I)
11    CONTINUE
C
C
C   ****************************************************************************
C
C     DATA IN     SET UP FOR OUTSIDE LOOP
C
       ICNTR=0
198     CONTINUE
       ICNTR=ICNTR+1
C
CIF  THERE ARE NO MORE VALUES. GO TO THE GEGINNING FOR MORE DATA
C
      IF(ICNTR.GT.6)GOTO 300
      IF(RLVEC(ICNTR).EQ.0.)GOTO 300
       LR=IFIX(RLVEC(ICNTR))
      FMORT=FSTRT-FINCR
200    FMORT=FMORT+FINCR
      IF(FMORT.GT.FSTOP)GOTO 198
C
C      SET UP FOR LOOP THROUGH YEARS (MIDDLE LOOP)
C
      YW=0.
       YN=1.E-10
      PW=0.
      PN=0.
       DO 22 I=1,NL
       EGGERS(I)=0.
       HOLD(I)=0.
22    POP(I)=PINIT(I)
      AL=0.
      YAL=0.
      YAW=0.
      TSPAWN=0.
       WRITE(6,16)FMORT,LR
C
C
```

```
C
C  *************************************************************************
C
C         12'     SET UP FOR LOOP THROUGH LENGTHS (INSIDE LOOP)
      SUMN=0.
      SUMW=0.
      CSUMN=0.
      CSUMW=0.
       PAL=0.
      CAL=0.
      SPAWN=0.
C     WORKING PART OF PROGRAM  AMORT IS A SURVIVORSHIP, APPLIED TO MOULTERS.
C     HOLD IS THE NUMBER MOLTING AT EACH LENGTH.
C     EGGERS  ARE FEMALES THAT DIDNT MOULT AND WILL BE BERRIED DURING
C     THE  NEXT FISHING SEASON, FEMALES EITHER MOULT OR BERRY-UP.
C
C     FROM LARGEST  TO SMALLEST
      DO 30  IK=1,NL
      I=(NL-IK)+1
      IF(POP(I)+EGGERS(I).LE.1.E-8) GO TO 29
C     CALCULATE THE CATCH, SUBTRACT IT FROM FREE ANIMALS
      CAT(I)=POP(I)*FMORT
      IF(LS+I.LT.LR)CAT(I)=0.
      POP(I)=POP(I)-CAT(I)
C     PUT ASIDE THE ANIMALS THAT WILL MOULT, FOR DISTRIBUTION INTO LARGER
C     LENGTH CATEGORIES, THESE ANIMALS WILL BE AVAILABLE FOR THE NEXT
C     FISHERY.
      HOLD(I)=PRMOLT(I)*POP(I)
      POP(I)=(POP(I)-HOLD(I))*BMORT
C     FEMALES THAT DONT MOULT GET BERRIED AND MISS THE FISHING MORTALITY
C     FOR ONLY THE NEXT FISHERY. THEY MOULT AFTER SPAWNING, AND ARE OPEN
C     FOR FISHING IN THAT NEXT YEAR.
      IF(ISEX.EQ.1)GOTO 25
C
C     ADD LAST YEARS BERRIED TO THIS YEARS MOULTERS
      HOLD(I)=HOLD(I)+EGGERS(I)
C     THE RESIDUAL WHEN THIS YEARS MOULTERS ARE ACCOUNTED FOR ARE THE
C     FEMALES THAT WILL BERRY.
      EGGERS(I)=POP(I)
      POP(I)=0.
25    CONTINUE
C     AT THIS POINT, ALL ANIMALS ABOUT TO MOULT ARE IN  HOLD.  FEMALES
C     WITH EGGS ARE SET ASIDE FOR 1 CYCLE IN EGGERS, THE ANIMALS THAT
C     WERE IN EGGERS LAST YEAR WERE TRANSFERRED TO THIS YEARS PREMOULTERS
C
C     ACCOUNT FOR MORTALITY ASSOCIATED WITH MOULTING
      HOLD(I)=HOLD(I)*AMORT
C
C     MOULTING AND  GROWTH
C
C     DISTRIBUTE  GROWERS AROUND PREDICTED  POSTMOULT LENGTH ACCORDING
C     TO A SUPPLIED  DISTRIBUTION TABLE, 10 SURROUNDING CLASSES USED.
      IZ=(IGROW(I)-LS)-6
      DO 32  K=1,10
      IX=IZ+K
C     CHECK THAT THIS IS A LEGAL SIZE TO GROW INTO.
      IF(IX.LT.1)IX=1
      IF(IX.GT.NL)IX=NL
      POP(IX)=POP(IX)+GDIST(K)*HOLD(I)
32    CONTINUE
       GO TO 30
29     CAT(I)=0.
       HOLD(I)=0.
      EGGERS(I)=0.
30    CONTINUE
C
C     WRITE POPULATION OUT, AND DO SUMS.
C
      DO  31  I=1,NL
      IF(POP(I)+HOLD(I)+EGGERS(I).LE.0.)GOTO 31
      SUMN=SUMN+POP(I)
       SUMW=SUMW+(POP(I)+EGGERS(I))*W(I)
      A=LS+I
      PAL=PAL+(POP(I)+EGGERS(I))*A
      CSUMN=CSUMN+CAT(I)
      CSUMW=CSUMW+CAT(I)*W(I)
      SPAWN=SPAWN+EGGERS(I)
31    CAL=CAL+CAT(I)*A
      AW=0.
      IF(SUMN.GT.0.)AW=SUMW/(SUMN+SPAWN)
      A=0.
      IF(SUMN.GT.0.)A=PAL/(SUMN+SPAWN)
      CAW=0.
      IF(CSUMN .GT.0.)CAW=CSUMW/CSUMN
      IF(CSUMN.GT.0.)CAL=CAL/CSUMN
```

-13-

```
      AL=AL+PAL
      YW=YW+CSUMW
      YN=YN+CSUMN
      PW=PW+SUMW
      PN=PN+SUMN
      YAL=YAL+CAL*CSUMN
      TSPAWN=TSPAWN+SPAWN
C
C
C
C     CHECK TO SEE IF ANOTHER YEAR IS GOING TO CHANGE THESE VALUES.
C
      IF(SUMN.LT.STARTN * .001)GOTO 101
100   CONTINUE
      WRITE(6,17)
17     FORMAT('    PROGRAM  WENT BEYOND ALLOTTED NO. OF YEARS')
101   CONTINUE
      A=YAL/YN
      B=YW/YN
      C=AL/PN
      D=PW/PN
      YN=YN/STARTN
      YW=YW/STARTN
      PW=PW/STARTN
      PN=PN/STARTN
      TSPAWN=TSPAWN/STARTN
      WRITE(6,8)YN,YW,A,B,PN,PW,C,D
      WRITE(6,9)TSPAWN
C
C     GO BACK TO THE START OF THE OUTSIDE LOOPS FOR NEW VALUES OF
C         FISHING AND  RECRUIT LENGTH
      GO TO  200
C
C
C  ******************************************************************************
C
C                **********       FORMAT STATEMENTS      **************
1     FORMAT(10F8.0)
2     FORMAT(20A4)
4     FORMAT('0LENGTH(MM)  INIT.POP.   PROP.MOULTING   POSTMOULT LENGTH
     +  WEIGHT (GRAMS)'/)
3     FORMAT(4X,I4,5X,  F10.0,F12.4,8X,I10,7X,F9.2)
8     FORMAT('0YIELD PER RECRUIT VALUES:'//
     +' YIELD IN NUMBERS:',F10.4/
     +' YIELD IN WEIGHT :',F10.2/
     +' YIELD MEAN LENGTH:',F10.2/
     +' YIELD MEAN WEIGHT:',F10.2/
     +' POPULATION NUMBERS:',F10.4/
     +' POPULATION WEIGHT:', F11.2/
     +' POPULATION MEAN LENGTH:',F10.2/
     +' POPULATION MEAN WEIGHT:',F10.2)
9     FORMAT(' NUMBER OF BERRIED FEMALES:',F10.2)
12    FORMAT('1',20X,20A4/)
13    FORMAT('0STARTING LENGTH:',I5//' LAST LENGTH:',I5/' NO. OF LENGTHS
     +:',I5/' MOULTING MORTALITY:',F8.3/' HARD-SHELLED MORTALITY:',F8.3/
     +' A AND B OF LN(W)=A+B*LN(L):',2F10.5/
     +' A AND B OF POSTMOULTLENGTH=A+B*PREMOULTLENGTH:',2F10.5)
14    FORMAT(' PROP. OF GROWERS AROUND PREDICTED LENGTH,BY MILLIMETERS:'
     +/10X,10F8.4)
15     FORMAT(' CONTROL VALUES FOR FISING MORTALITY AND RECRUIT LENGTH'
     +/10X,'FISH. MORT.  START, STOP, INCREMENT:',3F10.2/
     +10X,' REC. LENGTHS INPUT:',6F8.0)
 16   FORMAT(//' EXPLOITATION RATE',F6.2,5X,'RECRUITING LENGTH',I5/)
      END
/*
//GO.SYSIN DD *

/*
//
```